

S P O O L S



**THE
P O O L S
F O R E C A S T I N G
S Y S T E M
F O R
4 8 K S P E C T R U M**

SPOOLS

COPYRIGHT DR. M.A. BOURNE

The program Spools is copyright and is sold subject to the condition that it shall not be lent, resold, hired out or otherwise circulated.

DR. M.A. BOURNE
31 ARUNDEL AVE.
LIVERPOOL L17 3BY

NOTICE

All efforts have been made so that this manual and the Spools program are free from errors. M.A. Bourne can assume no responsibility for any errors or their consequences, in this manual or the Spools program.

SPOOLS

CONTENTS	PAGE
INTRODUCTION	1-2
GETTING STARTED	3
HOW SPOOLS WORKS - THE DETAILS -	4-9
REM data access and storage	4
The use of REM data storage in Spools	5
The menu and the first two menu items	5
Menu items three and four	6
Menu item five	7
How to use the score draw ratings	7
Menu items six to nine	7
Putting in new parameters (criteria/ideas)	8
Subroutines and menu items ten to thirteen	8-9
What can I do next season?	9
Possibilities, Databases and Microdrives	9
THE LISTING	10-

INTRODUCTION.

For many years now, the football pools have provided people with a challenge. This challenge is of course to predict as many of next weeks score draws as possible. Apart from the obvious intellectual satisfaction of success, thousands of people seem willing to chip in to reward the successful punter.

The employees and agents of pools companies who administer this transaction generally have families to feed, inevitable taxes and like to go on holidays now and again. So of course they need their "share of the action". This is generally in the region of one third of the monies tendered.

So where does this leave the punter? Basically in competition with all the other punters, success depending on the tools that one has available.

Let us consider a few of these tools;-

- (1)-Luck (everyone needs a bit of this)
- (2)-Intuition combined with a knowledge of football.
- (3)-A newspaper with expert and or computer forecasts.
- (4)-The personal computer.

LUCK

Let us go to randomland (which can, with a little thought be modelled on your computer).

In randomland there are 100 people and a pools company. Each week, each person fills in 3 pounds worth of perms on their coupons. So the company takes 300 pounds each week. 100 pounds is needed in overheads. Leaving 200 pounds which is split into two dividends, and given to the punters with the best lines according to the results of the random league.

How will the people get on, let us say after an average 100 week season (They have a funny calendar!);-

13 of the people will have won no dividends (lost 300 pounds)
 27 of the people will have won one dividend (lost 200 pounds)
 28 of the people will have won two dividends (lost 100 pounds)
 18 of the people will have won three dividends (broken even)
 9 of the people will have won four dividends (100 pounds profit)

5 of the people will have won five or more dividends (200 pounds + in profit)

So in randomland only 14% of the population can really call themselves lucky. The reward for their luck is not spectacular considering the investment.

This model is not meant to accurately reflect what goes on in the real world, but merely makes the point that in the long run, chance alone is no aid if the odds are against you.

To hammer this point further home;- after a 2000 week season in randomland NOBODY AT ALL is likely to have made a profit and losses will average 2000 pounds per person!

So much for luck, lets take a look at;-

INTUITION + KNOWLEDGE OF FOOTBALL

By intuition, I mean in this context, the skill of making reasonable guesses based on experience of what will predict a score draw. These guesses can then be applied on successive weeks to see if they provide any reliable guide to form.

This can be worked the other way around, looking at score draws and finding what seperates them from other fixtures.

These methods are very useful and can provide a start in narrowing the odds against a win. But there are 55 fixtures per week to process and if the punter is using several criteria to predict the draws then the task of sorting them out becomes rather daunting.

Knowledge of current events in football is also useful- if you have it (I don't, if you asked me what effect city's new sweeper is likely to have on goal averages I would probably stare blankly). If you know, while you fill in your coupon that 1/2 the city team has contracted AIDS -then good on you- make the most of all information available.

NEWSPAPER TIPS

Newspapers, with their expert tipsters and computers, vary in quality and consistency. But it seems to me that the better they are, the more people will base their coupons upon them -thus lowering the dividends if they happen to be right.

Now for my plug about

PERSONAL COMPUTERS

and their resourceful owners.

Computers are tools which, apart from creating aliens to !ZAP! have the ability to store everyday information. With the aid of programs, this information can be sorted, compared, updated and many other operations performed upon it which would otherwise be tedious to the human operator.

The football leagues contain just a few thousand pieces of info and so easily fit inside the 48K Spectrum with plenty of room for other information and a very substantial program to manipulate it.

A program in BASIC is (hopefully) flexible, understandable and accessible to the user. basic is however slow, but on the other hand, seconds and minutes are "cheaper" for home users than they are for multi national companies.

Most punters have their own ideas about what will predict a score draw. The Spools program will, (after a little practice) allow the punter to try out many of these ideas. Over the weeks feed back from the program will show which of the ideas are sensitive and specific indicators of score draws. Successful ideas can then be incorporated into the main body of the forecasting system.

As the system grows the user can be more and more confident that forecasts will be unique.

As I have hinted before, individuality in a pools forecasting system is as important as quality.

Quality is thus dependent to a certain extent upon the user- but individuality is assured. (That's not said very often in the computer age!)

So what will the diligent Spools user do with the weekly forecast? (!?) Fill in a coupon perhaps? Hopefully with the aid of any other "tools" at hand;-

Newspapers
 Knowledge of football
 Intuition
 and GOOD LUCK!

HOW TO WORK SPOOLS
AND
HOW SPOOLS WORKS

GETTING STARTED.

1. read the text up until the listing.
2. Load "Spools" into your 48K Spectrum and RUN.
3. Enter the date, then Answer; AUTO Y OR N ? with n
4. Answer; HARDCOPY Y OR N with n (unless printer is connected)
5. You will then be presented with the menu. Take your pick and mess around for half an hour getting to know the way info. is entered into spools and the sort of things the program does with this information.
6. You will not have to enter the full team names: each team has a code number (see list at back).
7. Pull the plug out and start again.
8. a) If you have the results of the matches for the date on the cassette and you have next weeks fixtures, then select from the menu in the following order; 1. 2. 3. 4. 5. 6. 12.
b) If you do not have those results and the program is just one or two weeks out of date then don't worry. Get this Saturday's fixtures and run the program with items from the menu in the following order; 2. 4. 5. 6. 12. You may at some stage wish to bring the leagues right up to date by going through menu item eight. But this is mindbendingly tedious if you do it all at once and is best done gently - say one or two divisions per week. N.B You will need the full 12 figure league tables to do this i.e. the ones which give details of home and away results.
c) If the leagues are hopelessly out of date, it may be worthwhile to start off by selecting item 8. (though you don't have to) Enter as many teams as you can bear, then menu items 2. 4. 5. 6. and 12. Then next week select menu item eight, and select items 1. 2. 3. 4. 5. 6. and 12. as per normal.

My feeling is that it is best to run the program as a Saturday only League, ignoring midweek matches. This has three advantages

- a) Much less hassle.
- b) Leagues will reflect Saturday matches which are the ones on the coupon.
- c) The increased individuality of forecasts will increase the size of dividends when they are right.

9. AFTER you have saved the program each week DO mess around with the program resident on the machine.

DO get to know how it works, especially the parts which do the predicting and test the criteria.

NEVER put in program lines less than 330 or you will corrupt the data.

ONCE you have thought of a criterion of your own, tested it and incorporated it into the program you will be justified in claiming that you produce your own INDIVIDUAL COMPUTER FORECASTS

HOW SPOOLS WORKS -THE DETAILS.

0-330 'REM' DATA ACCESS AND STORAGE.

A) ACCESS;- The data in the Spools program I.E. the leagues, the short term form, the fixtures, the results and a few other bits and peices are stored in REM statements.

Perhaps the best way of explaining this is with examples.

Try this on an empty Spectrum;

```
1 REM CAT*
2 REM DOG"
3 REM ANTI
10 FOR t = 0 TO 2 : LET s = 23760 + (t * 10)
20 FOR j = 0 TO 2
30 PRINT CHR$( PEEK (s + j ));
40 NEXT j
50 PRINT,
60 PRINT PEEK ( s + 3 )-33;"...LIVES"
70 NEXT t : STOP
```

23760 is the address in RAM of the first item after the first statement of the first line of any Spectrum program unless Micro drives are connected. In this case the number is the address of the letter 'c' of CAT. The j loop prints these letters. The code of the character '*' is 42 take off 33 leaves 9 which is the number of lives of a cat.

We add 33 to all numbers to be poked into the data area to avoid wierd errors - 13 for example would be interpreted as 'ENTER' and cause chaos. The number 42 (cats lives +33) is stored 3 bytes further on than the 'C' of CAT.

In the above program 4 pieces of information are stored using 4 bytes. An extra six bytes per line are necessary for the computer to register the line number, the REM statement itself and the end of the line.

In the Spools program there is room for up to 40 peices of information per line, with the necessary 6 bytes per line increment, this means an average of c. 1.125 bytes per peice of information. This is much more economical than arrays.

b) STORAGE;- Make the following additions/alterations to the CAT-DOG-ANT program;


```
14 INPUT "ENTER NAME OF ANIMAL ";n$
16 PRINT N$( 1 TO 3)
30 POKE s + j, CODE N$( j + 1)
50 INPUT " ENTER NUMBER OF LIVES ";l
60 PRINT l
65 POKE s+3, l
```

Run the altered program using names more than three letters and numbers of lives less than 222.

This kind of system is ideal for the football leagues where all numbers are integars well under 222. Spools uses 10 letter names for the teams, though the user refers to the teams with a number.

Now load the Spools program into your 48K Spectrum.
 The leagues and short term form are stored in REM lines 101-230.
 Last week's fixtures, this week's fixtures, the score, the
 result, the predicted ratings for results and whether the three
 tested criteria were satisfied are stored in lines 250-325.
 Lines 326-330 would be ideal sites for any machine code routines.

ANALYSIS OF TYPICAL REM STORAGE LINES.
 DON'T EXPECT TO UNDERSTAND THIS STRAIGHT AWAY, BE PATIENT.



800-999 THE BEGINNING OF THE PROGRAM.

800-899 Firstly dimensions all the arrays and prints the logo,
 then two input routines ask whether you require automatic mode
 and hardcopy.
 900-999 Contains the menu.

1000-1999 MENU ITEM ONE, INPUT LAST WEEK'S SCORES.

This routine takes each of last week's fixtures in turn and
 prints the team names (subroutine 9500). Address s contains the
 code number of the home team. Address s+1 contains the code
 number of the away team. The keyboard is then scanned for
 numbers between 0 & 9 (home score), short beep, then scanned
 again for the away score. After another short beep the scores
 plus 33 are poked into s+5 and s+6 respectively.
 1110-1160 is concerned with working out whether the scores
 represent home or away win, draw, score draw or no result (9-9)
 and poking the answer into s+7.

TWO POINTS 1. Don't let your fingers linger on the keyboard.
 2. If you do make a mistake, start again (sorry)
 with BREAK GOTO 1000 ENTER.

2000-2999 MENU ITEM TWO, INPUT NEXT WEEK'S FIXTURES.

In this rather simpler routine, you are invited to enter the
 codes of the teams for the next week's fixtures. The team names
 are printed and the codes plus 33 are poked into ram locations
 s and s+1. Note that the variable s in this routine always has a
 value 2 higher than in the score input routine so last week's
 fixtures are not overwritten. (Until menu item 6).

If you make a mistake, eg. enter the wrong team number, then
 note the fixture and when it comes to inputting the result next
 week, key 9 9 (no result).

3000-3999 MENU ITEM THREE, UPDATE LEAGUE DIVISIONS.

 This is a complex operation (for the computer, you don't do anything!). It contains within it several routines used by other menu items.

This is a brief summary in English, rather than Basic, of what the routine does:

3000-3090 For each of last week's fixtures; print the team names short term form and league information. Put the short term form and league information into arrays.

3100-3120 Deals with the increments which occur after any match, irrespective of result.

3125-3160 Goto the appropriate updating routine depending on whether the result was; no result, home win, away win, draw or score draw.

3200-3400 Updating routines.

N.B. can you spot those lines which sort out whether or not the fixture was Scottish? (requiring only 2 points for a win)

3500 Print the score.

3505-3620 Poke updated short term form and league information back into RAM and print the updated information.

4000-4999 MENU ITEM FOUR, MAKE PREDICTIONS.

 This routine is also rather complex but you will need to understand it if and when you want to put in your own ideas.

Again, a brief summary in English;

4000 For each fixture, 4020 use 3010-3085 to put the team data into arrays and print them.

4030 set the score draw rating at zero.

4032-4038 Prevent "number to big errors" by not allowing potential denominators to equal zero.

4040 If the away team has between 25% and 65% more points than the home team then increase the score draw rating by 2.

4050 If more than 15% of the home teams total games are home draws then increase the rating by one.

4060 If more than 15% of the away teams total games are away draws then increase the rating by one.

4070 If the home teams last three home games were draw, draw, draw (DDD) or were (WDD) or (LDD) or (DLD) then increase the rating by one.

4080 If the away teams last three away games were draw, draw, draw (DDD) or (WDD) or (LDD) or (DWD) then increase the rating by one.

4081-4189 Space for you to put in your own ideas for predicting score draws.

4190 Print score draw rating and store it for future reference.

4200-4390 Routine for producing a home win rating (you can put in your own criteria here too!).

4400-4590 Routine for producing an away win rating (and here!).

4600 Find whether first criterion to be tested is satisfied.

4610 Remember that.

4650-60 Second criterion.

4700-10 Third criterion.

The above information (4600-4710) is used next week, once the results are in, in the "test criterion" section to tell us how good the criteria were at predicting score draws.

4800-4840 Print whether or not each criterion was satisfied for the fixture.

4850-4910 Go BEEP, then get the next fixture unless we've finished predicting. If in automatic mode go on to display summary of predictions, otherwise go back to menu.

5000-5999 MENU ITEM FIVE. PRINT SUMMARY OF PREDICTIONS.

This routine prints a summary of the predictions for each fixture. If you have a printer connected and have asked for hardcopy we GOSUB 5500, print the fixtures and three numbers by each fixture. These are the score draw, home win and away win ratings respectively.

The same is then done on screen for those without benefit of printers.

HOW TO USE THE RATINGS.

Once we have our printout, or have the predictions on screen, how do we interpret them and what do we do with them?

Note that the highest numbers in the SD column are higher than those in the HW and AW columns, This is because fewer criteria are used to sort out the obvious home and away bankers.

Firstly we eliminate all the fixtures which are obvious home or away bankers, and those which the program could come to no clear decision over.

Now, commence picking fixtures, starting with ones with high SD ratings and zero HW and AW ratings, and finishing with those that have less favourable ratings, until you have say 12-18 fixtures in order.

You can now select from these to fill your coupon in your usual way. Don't forget to use your brain and all the other tools available to the punter. For instance, you may wish to eliminate those fixtures which are tipped in more than one newspaper, as if they are score draws they may tend to lower dividends.

6000-6999 MENU ITEM SIX. TRANSFER FIXTURES FOR NEXT WEEK.

When last week's scores have been entered, the leagues updated and we have put in next week's fixtures, we have no further need for last week's and will want next week's fixtures to be ready to have their scores entered next week.

This routine arranges this by poking next week's fixtures into the positions occupied by last week's - overwriting them.

7000-7999 MENU ITEM SEVEN. YOUR ROUTINES.

Simply some space in the listing for you to put in your own routines.

8000-8999 MENU ITEM EIGHT. INPUT THE DIVISIONS.

This little routine allows you to enter full league information team by team.

As I've hinted before, this gets rather tedious after a while and is best done bit by bit.

9000-9380 MENU ITEM NINE. TEST CRITERIA.

Remember that 4600-4710 remembered for each fixture whether each of three criteria for predicting score draws were satisfied. By comparing this information with the actual scores for each fixture, we can get some idea of how good these criteria are.

For each fixture;-
 i will be ↑ by one if the result was a score draw.
 j(1) will be ↑ by 1 if crit 1 predicted a score draw.
 j(2) will be ↑ by 1 if crit 2 predicted a score draw.
 j(3) will be ↑ by 1 if crit 3 predicted a score draw.
 k(1) ↑ by 1 if score draw and crit 1 predicted a score draw.
 k(2) ↑ by 1 if score draw and crit 2 predicted a score draw.
 k(3) ↑ by 1 if score draw and crit 3 predicted a score draw.
 Once the cumulative totals for all the fixtures have been found the criteria are printed (by list 9997, so if you change a criterion, remember to change the appropriate line at the end, 9997, 9998, or 9999.). We then print the number of score draws (i), and for each criterion, the number of score draws correctly and wrongly forecast.
 If hardcopy is required, a copy is made.

WHAT IF I WANT TO TEST ANOTHER IDEA?

Criterion 2 is not very specific and picks out too many of the fixtures as potential score draws.
 As it stands it picks a fixture if more than 15% of all the home team's games were home draws. To make it more selective we would want to increase this to say, 22% How do we do this?
 Answer: Change the "0.3" in line 4650 to "0.44".
 and do the same in line 9998.
 If it works reasonably over a few weeks, change line 4050 as well.

EXERCISES:

Is the criterion in line 4060 specific enough?
 Can I improve it in the same way?
 How could I put in a criterion to randomly pick out about 8 fixtures a week so I can compare the other criteria with chance?

9500 PRINT TEAM NAME SUBROUTINE.

This single line subroutine is used whenever the program requires a team name to be printed.
 Note 23760 is the address of the A in Arsenal.
 23760 + 46 = 23806 which is the address of the A in Aston villa

9550 LPRINT TEAM NAME SUBROUTINE.

As above with LPRINT.

9600-9630 MENU ITEM ELEVEN. PRINT LEAGUES.

This routine asks for the range of teams that you want to see. It prints the team numbers, names, short term form and league information for that range of teams.

9650-9690 MENU ITEM TEN. HARDCOPY LEAGUES.

This does the same as the above routine, but output goes to the printer rather than the screen.
 Note, if you have an 80 column printer, you can improve the format of the printing to produce output similar to normal league tables.

9700-9720 MENU ITEM TWELVE. SAVE SPOOLS.

Not too difficult, always verify the program but please do not sell or give versions to anyone.

MENU ITEM THIRTEEN. INTO BASIC.

Acheived simply by line 993.

WHAT DO I DO NEXT SEASON?

By the end of the season I hope you will be broadly familiar with the working of the program.

You will need to change team numbers to accommodate promotion and relegation. This can be done by crafty editing of the line numbers.

I recomend that you leave the short term form as it stands.

You may wish to wipe the divisions clean with the following routine:

```
7000 FOR a = 1 TO 131 : GOSUB 9500
7010 FOR j = 17 TO 28 : POKE b + j,33 : NEXT j
7020 CLS : NEXT a : GOTO 900
```

POSSIBILITIES FOR EXPANSION.

1. DATABASES.

You could create your own data base by making a block of info each week containing fixtures and scores and saving it on tape. Alternatively, why not use one of the databases already on the market, in conjunction with Spools.

2. MICRODRIVES.

Unfortunately, when Microdrives are connected, the address of the beginning of basic programs will vary, causing problems with the Spools addressing system.

To overcome this you will need to poke all the data high up in RAM and change all the values of variables s and b appropriately.

Or more extensive surgery could be used to employ the data streams of the Microdrives.

Anyway, if Microdrives actually exist, and become widely available before the end of the century, the possibilities for the Spools program will increase enormously.

Remember, the Spools program is written in basic, the language of the home user - so that you can have control of its potential to produce unique forecasts.

1SPOOLS LISTING

```
101 REM ARSENAL 1LLWQLW'##('!"%*aaaaaaaaaaaa
102 REM ASTON VILL1WWWLWL'##'##'+aaaaaaaaaaaa
103 REM BIRMINGHAM1WWWLDD(##!%!!##),aaaaaaaaaaaa
104 REM COVENTRY 1DWWLL'##!&##!#&)+aaaaaaaaaaaa
```

etcetera until

```
229>REM STIRLING A7WWWQLD&#!&#!!"%&<aaaaaaaaaaaa
230 REM STRANRAER 7WDLWL'##'##'##'##<aaaaaaaaaaaa
231 REM XXXXXXXXXXXX.aaaaaaaaaaaaaaaaaaaaaaaaaaaa
250 REM $(($ $!"!&!!!!)aaaaaaaaaaaaaaaaaaaa
251 REM %'%'a"!"$!"!"!aaaaaaaaaaaaaaaaaaaa
252 REM )2)2a"%"!$!!!!)aaaaaaaaaaaaaaaaaaaa
253 REM *##*a$%#"#!!!!)aaaaaaaaaaaaaaaaaaaa
```

etcetera until

```
329>REM aaaaaaaaaaaaaaaaaaaaaaaaaaaaaaaaaaaaaa
330 REM aaaaaaaaaaaaaaaaaaaaaaaaaaaaaaaaaaaaaa
800 DIM E$(2,3): DIM G$(2,3): DIM N$(2,1): DIM F(12): DIM H(12): DIM T(3): DIM
J(3): DIM K(3)
```

```
805 LET a$=""
810 FOR x=1 TO 44: PRINT INK 7: PAPER 1:a$: NEXT x
830 PAPER 0: BORDER 2: INK 7: POKE 23609,60
840 PRINT AT 1,2:" SPOOLS POOLS FORECASTING AID."
850 PRINT AT 3,2:" @ DR.M.A.BOURNE."
860 PRINT AT 5,2:" 31 ARUNDEL AVE.LIVERPOOL 17": PRINT
865 INPUT "ENTER the date ":f$
867 PRINT AT 11,10:" ":f$:" ": COPY
870 INPUT "AUTO Y OR N ":d$
880 INPUT "HARDCOPY Y OR N ":p$
900 CLS
901 PRINT "-1- INPUT LAST WEEKS SCORES"
905 PRINT "-2- INPUT NEXT WEEKS FIXTURES"
910 PRINT "-3- UPDATE LEAGUE DIVISIONS"
915 PRINT "-4- MAKE PREDICTIONS"
920 PRINT "-5- PRINT SUMMARY OF PREDICTIONS"
925 PRINT "-6- TRANSFER FIXTURES FOR NEXT WEEK"
930 PRINT "-7- SUB ROUTINES"
935 PRINT "-8- INPUT THE DIVISIONS"
940 PRINT "-9- TEST CRITERION"
945 PRINT "-10-HARDCOPY LEAGUES"
950 PRINT "-11-PRINT LEAGUES"
955 PRINT "-12-SAVE SPOOLS"
960 PRINT "-13-INTO BASIC"
980 IF d$="y" THEN GO TO 1000
985 PRINT : PRINT FLASH 1:"MAKE YOUR SELECTION"
990 INPUT a: IF a=10 THEN CLS : GO TO 9650
991 IF a=11 THEN CLS : GO TO 9600
992 IF a=12 THEN CLS : GO TO 9700
993 IF a=13 THEN GO TO 10000
995 CLS : GO TO a*1000
```

```

1000>CLS :REM Score Input
1010 PRINT ; INVERSE 1;"INPUT LAST WEEKS SCORES. FOR      GAMES WITH UNKNOWN SCORE
S E.G. POSTPONED INPUT THE SCORE AS  9 9 AND PUT FIXTURE IN AT A      LATER DATE.
"
1015 PRINT  FLASH 1;" DON'T LET YOUR FINGERS LINGER! "
1020 PRINT : LET s=29786: FOR t=1 TO 75: PRINT t;TAB 3;
1025 POKE 23692,255
1030 LET a=PEEK (s)-33: GO SUB 9500: PRINT , " "
1040 LET a=PEEK (s+1)-33: GO SUB 9500: PRINT
1050 IF PEEK (s)=164 THEN  GO TO 1170
1060 FOR J=1 TO 2
1070 LET s%=INKEY$: IF NOT (CODE s%)=48 AND CODE s%(<=57) THEN  GO TO 1070
1075 BEEP 0.6,(J-1)*12
1080 LET n%(J)=s%: PRINT TAB J*14; PAPER 0; INK 5;n%(J);
1090 POKE s+4+J,CODE n%(J)-15
1100 NEXT J: PRINT
1110 IF PEEK (s+5)=42 AND PEEK (s+6)=42 THEN  PRINT "no result": GO TO 1160
1120 IF PEEK (s+5)>PEEK (s+6) THEN  PRINT "home win": POKE s+7,34
1130 IF PEEK (s+5)<PEEK (s+6) THEN  PRINT "away win": POKE s+7,35
1140 IF PEEK (s+5)=33 AND PEEK (s+6)=33 THEN  PRINT "scoreless draw": POKE s+7,3
6
1150 IF PEEK (s+5)=PEEK (s+6) AND PEEK (s+5)>33 THEN  PRINT "score draw": POKE s
+7,37
1160 LET s=s+36: NEXT t
1170 INPUT "Tester y or n ? ";a$: IF a$="y" THEN  GO TO 9000
1180 IF NOT d$="y" THEN  GO TO 900
2
2000>CLS :REM Input next weeks fixtures
2010 PRINT ; INVERSE 1;"ENTER NEXT WEEKS FIXTURES E.G.  12<enter>14<enter>": PRI
NT "AFTER THE LAST FIXTURE ENTER 131 TWICE": PRINT
2020 LET s=29786: LET n=1
2030 PRINT "ENTER FIXTURE ";n: PRINT "home team";
2040 INPUT a: PRINT " ";a;" ";
2045 GO SUB 9500: PRINT
2047 IF a>=132 THEN  PRINT " NUMBER TOO BIG, TRY AGAIN": BEEP 2,0: GO TO 2030
2050 POKE s,a+33
2055 INVERSE 1: PRINT "away team";
2060 INPUT a: PRINT " ";a;" ";: GO SUB 9500: INVERSE 0: PRINT : PRINT
2065 IF a>=132 THEN  PRINT "NUMBER TOO BIG, TRY AGAIN": BEEP 2,0: GO TO 2055
2070 POKE s+1,a+33: IF a=131 THEN  GO TO 2100
2080 LET n=n+1: LET s=s+36: IF n>75 THEN  GO TO 2100
2090 GO TO 2030
2100 IF NOT d$="y" THEN  GO TO 900
3
3000>CLS :LET s=29786:LET pred=0:LET nine=0:LET t=0
3010 CLS : PRINT TAB 2;t+1;" ";: PRINT "home team"
3020 LET a=PEEK (s)-33: GO SUB 9500: PRINT " ";
3030 FOR J=1 TO 2: LET e%(J)=CHR$ (PEEK (b+11+((J-1)*3)))+CHR$ (PEEK (b+12+((J-1
)*3)))+CHR$ (PEEK (b+13+((J-1)*3))): PRINT e%(J);" ";: NEXT J: PRINT
3040 FOR J=1 TO 12: LET f%(J)=PEEK (b+16+J)-33: PRINT f%(J);" ";: NEXT J: PRINT :
PRINT
3050 LET a=PEEK (s+1)-33: PRINT TAB 4;"away team ": GO SUB 9500: PRINT " ";
3060 FOR J=1 TO 2: LET g%(J)=CHR$ (PEEK (b+11+((J-1)*3)))+CHR$ (PEEK (b+12+((J-1
)*3)))+CHR$ (PEEK (b+13+((J-1)*3))): PRINT g%(J);" ";: NEXT J: PRINT
3070 FOR J=1 TO 12: LET h%(J)=PEEK (b+16+J)-33: PRINT h%(J);" ";: NEXT J: PRINT :
PRINT : PRINT
3080 IF PEEK s)=164 AND PEEK (s+1)>=164 THEN  GO TO 3620
3085 IF pred=1 THEN  RETURN
3090 IF PEEK (s+5)=42 AND PEEK (s+6)=42 THEN  PRINT "no result": GO TO 3590
3100 LET f(1)=f(1)+1: LET h(1)=h(1)+1
3110 LET f(5)=f(5)+PEEK (s+5)-33: LET f(6)=f(6)+PEEK (s+6)-33
3120 LET h(10)=h(10)+PEEK (s+6)-33: LET h(11)=h(11)+PEEK (s+5)-33
3125 IF PEEK (s+5)=42 AND PEEK (s+6)=42 THEN  PRINT "no result": GO TO 3950
3130 IF PEEK (s+7)=34 THEN  GO SUB 3200
3140 IF PEEK (s+7)=35 THEN  GO SUB 3300
3150 IF PEEK (s+7)=36 THEN  GO SUB 3400
3160 IF PEEK (s+7)=37 THEN  GO SUB 3400
3170 GO TO 3500

```

```

3200 IF b<27992 THEN LET f(12)=f(12)+3
3210 IF b>=27992 THEN LET f(12)=f(12)+2
3220 LET f(2)=f(2)+1: LET h(9)=h(9)+1
3230 LET e$(1)=e$(1)(2 TO 3)+"W"
3240 LET g$(2)=g$(2)(2 TO 3)+"L"
3250 RETURN
3300 IF b<27992 THEN LET h(12)=h(12)+3
3310 IF b>=27992 THEN LET h(12)=h(12)+2
3320 LET f(4)=f(4)+1: LET h(7)=h(7)+1
3330 LET e$(1)=e$(1)(2 TO 3)+"L"
3340 LET g$(2)=g$(2)(2 TO 3)+"W"
3350 RETURN
3400 LET f(12)=f(12)+1: LET h(12)=h(12)+1
3410 LET f(3)=f(3)+1: LET h(8)=h(8)+1
3420 LET e$(1)=e$(1)(2 TO 3)+"D"
3430 LET g$(2)=g$(2)(2 TO 3)+"D"
3440 RETURN
3500 PRINT INVERSE 1;" SCORE ";PEEK (s+5)-33;" ";PEEK (s+6)-33;" "
3505 PRINT INVERSE 1;" AFTER UPDATE "": PRINT : LET a=PEEK (s
-33: GO SUB 9500: PRINT " ";
3510 LET c#=e$(1)+e$(2): PRINT c#
3520 FOR j=1 TO 6: POKE b+11+(j-1),CODE c$(j): NEXT j
3530 FOR j=1 TO 12: POKE b+16+j,f(j)+33: PRINT f(j);" "; NEXT j: PRINT : PRINT
3550 LET a=PEEK (s+1)-33: GO SUB 9500: PRINT " ";
3560 LET c#=g$(1)+g$(2): PRINT c#
3570 FOR j=1 TO 6: POKE b+11+(j-1),CODE c$(j): NEXT j
3580 FOR j=1 TO 12: POKE b+16+j,h(j)+33: PRINT h(j);" "; NEXT j: PRINT : BEEP 1
.0
3590 LET up=0: LET s=s+36: LET t=t+1: IF nine=1 THEN RETURN
3600 IF t<=75 THEN GO TO 3010
3610 IF pred=1 THEN GO TO 5000
3620 IF NOT d$="y" THEN GO TO 900
4
4000>CLS :REM PREDICTIONS
4010 LET pred=1: LET s=29798: LET t=0
4020 GO SUB 3010
4030 LET p=0
4032 FOR j=1 TO 12
4034 IF f(j)=0 THEN LET f(j)=1
4036 IF h(j)=0 THEN LET h(j)=1
4038 NEXT j
4040 IF h(12)/f(12)>1.25 AND h(12)/f(12)<1.65 THEN LET p=p+2
4050 IF f(3)*2/f(1)>0.3 THEN LET p=p+1
4060 IF h(8)*2/h(1)>0.3 THEN LET p=p+1
4070 IF e$(1)="DDD" OR e$(1)="WDD" OR e$(1)="LDD" OR e$(1)="OLD" THEN LET p=p+1
4080 IF g$(2)="DDD" OR g$(2)="WDD" OR g$(2)="LDD" OR g$(2)="DWD" THEN LET p=p+1
4190 PRINT "Score draw rating ";p: POKE s+6,(p+33)
4200 LET w=0
4210 IF f(12)/h(12)>=1 THEN LET w=w+1
4220 IF f(12)/h(12)>1.5 THEN LET w=w+2
4240 IF e$(1)="WWW" OR e$(1)="DWW" THEN LET w=w+1
4250 IF g$(2)="LLL" OR g$(2)="DLL" OR g$(2)="LDL" OR g$(2)="LLD" THEN LET w=w+1
4390 PRINT "Home win rating ";w: POKE s+7,(w+33)
4400 LET l=0
4410 IF h(12)/f(12)>1.8 THEN LET l=l+1
4420 IF h(12)/f(12)>2.3 THEN LET l=l+2
4440 IF e$(1)="LLL" OR e$(1)="DLL" OR e$(1)="LDL" OR e$(1)="LLD" THEN LET l=l+1
4450 IF g$(2)="WWW" OR g$(2)="DWW" OR g$(2)="WDW" OR g$(2)="WWD" THEN LET l=l+1
4590 PRINT "Away win rating ";l: POKE s+8,(l+33)

```

```

4600 LET t(1)=0: IF h(12)/f(12)>1.25 AND h(12)/f(12)<1.65 THEN LET t(1)=t(1)+1
4610 POKE s+9,(t(1)+33)
4650 LET t(2)=0: IF f(3)*2/f(1)>0.3 THEN LET t(2)=t(2)+1
4660 POKE s+10,(t(2)+33)
4700 LET t(3)=0: IF e$(1)="DDD" OR e$(1)="WDD" OR e$(1)="LDD" OR e$(1)="DLD" THEN
N LET t(3)=t(3)+1
4710 POKE s+11,(t(3)+33)
4800 FOR J=1 TO 3
4810 IF t(J)=1 THEN PRINT "Tested criterion "J;" Positive"
4820 IF t(J)=0 THEN PRINT "Tested criterion "J;" Negative"
4830 NEXT J
4840 PRINT
4850 BEEP 1,12
4900 LET s=s+36: LET t=t+1: IF t>=75 THEN GO TO 4920
4910 GO TO 4020
4920 IF NOT d$="y" THEN GO TO 900
5
5000>CLS :REM PRINT PREDICTIONS
5010 CLS : PRINT f$: LET s=29788: PRINT "PREDICTION SUMMARY          SD HW AW"
5015 IF p$="y" THEN GO SUB 5500
5020 LET s=29788: FOR t=1 TO 75
5030 IF PEEK s>=164 AND PEEK (s+1)>=164 THEN INPUT "Press ENTER to continue":r$
: GO TO 5100
5050 PRINT t;TAB 2;: LET a=PEEK s-33: GO SUB 9500: PRINT " ";: LET a=PEEK (s+1)-
33: GO SUB 9500
5060 PRINT TAB 24;PEEK (s+6)-33;TAB 27;PEEK (s+7)-33;TAB 30;PEEK (s+8)-33
5070 LET s=s+36
5080 NEXT t: PRINT
5090 INPUT "Press ENTER to continue":r$
5100 IF NOT d$="y" THEN GO TO 900
5110 GO TO 6000
5500 LPRINT f$: LPRINT "PREDICTION SUMMARY          SD HW AW"
5510 FOR t=1 TO 75
5520 IF PEEK s>=164 AND PEEK (s+1)>=164 THEN RETURN
5530 LPRINT t;TAB 2;: LET a=PEEK s-33: GO SUB 9550: LPRINT " ";: LET a=PEEK (s+1)
)-33: GO SUB 9550
5540 LPRINT TAB 24;PEEK (s+6)-33;TAB 27;PEEK (s+7)-33;TAB 30;PEEK (s+8)-33
5550 LET s=s+36
5560 NEXT t: LPRINT : RETURN
6
6000>CLS :PRINT FLASH 1;"TRANSFERRING FIXTURES"
6010 LET s=29786: FOR t=1 TO 75: POKE s,PEEK (s+2): POKE s+1,PEEK (s+3): IF PEEK
s=164 THEN CLS
6015 IF PEEK s=164 THEN GO TO 6100
6020 LET s=s+36: NEXT t: GO TO 900
6100 IF NOT d$="y" THEN GO TO 900
6110 PRINT INVERSE 1;"SPOOLS IS READY TO SAVE": GO TO 9700
7
7000>PRINT "THIS SPACE (7000-7999) RESERVED FOR YOUR ROUTINES":PAUSE 200
7005 GO TO 900
8
8000>REM LEAGUE INPUT
8010 PRINT INVERSE 1;"Enter the number of the team whose league info you wis
h to enter. Enter 999toescape."
8020 INPUT a: IF a=999 THEN GO TO 900
8025 PRINT a;TAB 3;
8030 GO SUB 9500
8050 FOR j=17 TO 28: INPUT z: PRINT TAB (j-17)*3;z;: POKE b+j,(z+33): NEXT j: PR
INT ' POINTS ':GOTO 8010

```



```

9000>REM TESTER
9010 LET i=0
9020 FOR n=1 TO 3: LET J(n)=0: LET k(n)=0: NEXT n
9030 PRINT "TESTER--Only accurate after the results have been entered but but
before next weeks Predictions have been made.": PRINT : INPUT "O.K? Y or N ";c$:
IF c$="n" THEN GO TO 1180
9050 CLS : PRINT FLASH 1;"RUNNING TESTER"
9060 LET s=29786
9070 FOR t=1 TO 75
9080 IF PEEK s>=164 OR PEEK (s+1)>=164 THEN CLS : GO TO 9200
9100 IF PEEK (s+7)=37 THEN LET i=i+1
9110 IF PEEK (s+11)=34 THEN LET J(1)=J(1)+1
9120 IF PEEK (s+12)=34 THEN LET J(2)=J(2)+1
9130 IF PEEK (s+13)=34 THEN LET J(3)=J(3)+1
9140 IF PEEK (s+7)=37 AND PEEK (s+11)=34 THEN LET k(1)=k(1)+1
9150 IF PEEK (s+7)=37 AND PEEK (s+12)=34 THEN LET k(2)=k(2)+1
9160 IF PEEK (s+7)=37 AND PEEK (s+13)=34 THEN LET k(3)=k(3)+1
9170 LET s=s+36
9180 NEXT t: CLS
9200 PRINT f$: LIST 9997
9300 PRINT "THERE WERE ";i;" SCORE DRAWS": PRINT
9310 FOR J=1 TO 3
9320 PRINT "CRIT ";J;" PREDICTED ";TAB 20;k(J);" OF THEM"
9330 PRINT "AND PREDICTED ";TAB 20;j(J)-k(J);" WRONGLY"
9340 PRINT
9350 NEXT J
9360 IF p$="y" THEN COPY
9365 INPUT "Press ENTER to continue ";r$
9370 IF d$="y" THEN GO TO 2000
9380 GO TO 900
9
9500>LET b=23760+(a*46)-46:FOR J=0TO 9:PRINT INK 6;PAPER 0;CHR$(PEEK (b+J));:NE
XT J:RETURN
9
9550>LET b=23760+(a*46)-46:FOR J=0TO 9:LPRINT CHR$(PEEK (b+J));:NEXT J:RETURN
9
9600>INPUT "ENTER THE RANGE OF TEAMS THAT YOU WANT PRINTED ";k;"TO";l
9610 FOR a=k TO l: POKE 23692,255: PRINT TAB 4;a;TAB 8;: GO SUB 9500: PRINT " ";
: FOR J=11 TO 16: PRINT CHR$(PEEK (b+J));: NEXT J: PRINT
9620 FOR J=17 TO 28: PRINT TAB (((J-17)*3));(PEEK (b+J))-33;: NEXT J: PRINT " PO
INTS": PRINT : NEXT a
9630 INPUT "PRESS ENTER TO CONTINUE ";r$: GO TO 900
9
9650>REM Hardcopy leagues
9660 INPUT "ENTER THE RANGE OF TEAMS THAT YOU WANT HARDCOPY OF";k;" TO ";l
9670 FOR a=k TO l: LPRINT TAB 4;a;TAB 8;: GO SUB 9550: LPRINT " ";: FOR J=0 TO
5: LPRINT CHR$(PEEK (b+11+J));: NEXT J: LPRINT
9680 FOR J=17 TO 28: LPRINT TAB (((J-17)*3));(PEEK (b+J))-33;: NEXT J: LPRINT "
POINTS": LPRINT : NEXT a
9690 INPUT "PRESS ENTER TO CONTINUE ";r$: GO TO 900
9
9700>PRINT "Disconect ear Plug from cass.":SAVE "spools"LINE 331:PRINT INVERSE 1
;"spools saved":PRINT :PRINT "Stop tape,rewind,reconnect ear Plug and play to
verify spools":VERIFY "spools"
9710 CLS : PRINT FLASH 1;"O.K. why not mess around a bit! "
9720 GO TO 901
9997>LET t(1)=0: IF h(12)/f(12)>1.25AND h(12)/f(12)<1.65THEN LET t(1)=t(1)+1
9998 LET t(2)=0: IF f(3)*2/f(1)>0.3 THEN LET t(2)=t(2)+1
9999 LET t(3)=0: IF e$(1)="DDD" OR e$(1)="WDD" OR e$(1)="LDD" OR e$(1)="DLD" THE
N LET t(3)=t(3)+1

```


CODES OF TEAM NAMES

DIVISION ONE

1 ARSENAL
2 ASTON VILL
3 BIRMINGHAM
4 COVENTRY
5 EVERTON
6 IPSWICH
7 LEICESTER
8 LIVERPOOL
9 LUTON
10 MAN UNITED
11 NORWICH
12 NOTTS CO
13 NOTTS FO
14 QPR
15 SOUTHAMPTO
16 STOKE
17 SUNDERLAND
18 TOTTENHAM
19 WATFORD
20 WEST BROM
21 WEST HAM
22 WOLVERHAMPTON

DIVISION THREE

45 BOLTON
46 BOURNEMOUTH
47 BURNLEY
48 BRADFORD
49 BRENTFORD
50 BRISTOL ROY
51 EXETER
52 GILLINGHAM
53 HULL
54 LINCOLN
55 MILLWALL
56 NEWPORT
57 ORIENT
58 OXFORD
59 PLYMOUTH
60 PORT VALE
61 PRESTON
62 ROTHERHAM
63 SCUNTHORPE
64 SHEFF UNITED
65 SOUTHBEND
66 WALSALL
67 WIGAN

DIVISION S.PREM

93 ABERDEEN
94 CELTIC
95 DUNDEE
96 DUND UNITED
97 HEARTS
98 HIBERNIAN
99 MOTHERWELL
100 RANGERS
101 ST. JOHNSTON
102 ST. MIRREN

DIVISION TWO

23 BARNLEY
24 BRIGHTON
25 BLACKBURN
26 CAMBRIDGE
27 CARDIFF
28 CARLISLE
29 CHARLTON
30 CHELSEA
31 CRYSTAL PA
32 DERBY
33 FULHAM
34 GRIMSBY
35 HUDDERSFIELD
36 LEEDS UNITED
37 MAN CITY
38 MIDDLESBROUGH
39 NEWCASTLE
40 OLDHAM
41 PORTSMOUTH
42 SHEFF WED
43 SHREWSBURY
44 SWANSEA

DIVISION FOUR

69 ALDERSHOT
70 BLACKPOOL
71 BRISTOL CITY
72 BURY
73 CHESTER
74 CHESTERFIELD
75 COLCHESTER
76 CREWE
77 DARLINGTON
78 DONCASTER
79 HALIFAX
80 HARTLEPOOL
81 HEREFORD
82 MANSFIELD
83 NORTHAMPTON
84 PETERBOROUGH
85 READING
86 ROCHDALE
87 STOCKPORT
88 SWINDON
89 TORQUAY
90 TRANMERE
91 WREXHAM

DIVISION S.ONE

103 AIRDRIE
104 ALLOA
105 Ayr
106 BRECHIN
107 CLYDE
108 CLYDEBANK
109 DUMBARTON
110 FALKIRK
111 HAMILTON
112 KILMARNOCK
113 MEADOWBANK
114 MORTON
115 PARTICK
116 RAITH

DIVISION S.TWO

117 ALBION
118 ARBROATH
119 BERWICK
120 COWDENBEAT
121 DUMFIRMLIN
122 E.FIFE
123 E.STIRLING
124 FORFAR
125 MONTROSE
126 QUEEN OF SCOTLAND
127 QUEENS PARK
128 STENHOUSEMOUTH
129 STIRLING A
130 STRANRAER